# Maintainer: Johannes Löthberg <johannes@kyriasis.com>
# Contributor: Ivan Shapovalov <intelfx@intelfx.name>

pkgname=python-unpaddedbase64
pkgver=2.1.0
pkgrel=5

pkgdesc='Unpadded Base64'
url='https://pypi.python.org/pypi/unpaddedbase64/'
arch=('any')
license=('Apache')

depends=('python')
makedepends=('python-build' 'python-installer' 'python-poetry-core')

source=("python-unpaddedbase64-${pkgver}.tar.gz::https://github.com/matrix-org/python-unpaddedbase64/archive/v$pkgver.tar.gz")

sha256sums=('8f3aeae62aad49ee385c15c37bb64b625c96bb85a820aeab149c21970843ba68')

build() {
	cd python-unpaddedbase64-$pkgver
	python -m build --wheel --no-isolation
}

check() {
	cd $pkgname-$pkgver
	python -m unittest discover -vs tests/
}
package() {
	cd python-unpaddedbase64-$pkgver
	python -m installer --destdir="$pkgdir" dist/*.whl
}

# vim: set ts=4 sw=4 tw=0 ft=PKGBUILD :
